The Semantic Web Rule Language


Martin O'Connor
Stanford Center for Biomedical Informatics Research,
Stanford University



Talk Outline

- Rules and the Semantic Web
- Basic SWRL Rules
- SWRL's Semantics
- SWRLTab: a Protégé-OWL development environment for SWRL
- SQWRL: a SWRL-based OWL query langauge

Semantic Web Stack



Rule-based Systems are common in many domains

- Engineering: Diagnosis rules
- Commerce: Business rules
- Law: Legal reasoning
- Medicine: Eligibility, Compliance
- Internet: Access authentication

Rule Markup (RuleML) Initiative

- Effort to standardize inference rules.
- RuleML is a markup language for publishing and sharing rule bases on the World Wide Web.
- Focus is on rule interoperation between industry standards.
- RuleML builds a hierarchy of rule sublanguages upon XML, RDF, and OWL, e.g., SWRL

What is SWRL?

- SWRL is an acronym for Semantic Web Rule Language.
- SWRL is intended to be the rule language of the Semantic Web.
- SWRL includes a high-level abstract syntax for Horn-like rules.
- All rules are expressed in terms of OWL concepts (classes, properties, individuals).
- Language FAQ:
 - http://protege.cim3.net/cgi-bin/wiki.pl?SWRLLanguageFAQ

SWRL Characteristics

- W3C Submission in 2004: http://www.w3.org/Submission/SWRL/
- Rules saved as part of ontology
- Increasing tool support: Bossam, R2ML, Hoolet, Pellet, KAON2, RacerPro, SWRLTab
- Can work with reasoners

Example SWRL Rule: Reclassification

 $Man(?m) \rightarrow Person(?m)$

Possible in OWL - some rules are OWL syntactic sugar

Example SWRL Rule: property value assignment

```
Person(?p) ^ hasSibling(?p,?s) ^ Man(?s)

→ hasBrother(?p,?s)
```

Example SWRL Rule: property value assignment

hasParent(?x, ?y) ^ hasBrother(?y, ?z) → hasUncle(?x, ?z)

Not possible in OWL 1.0 - some rules are *not* OWL syntactic sugar

Example SWRL Rule with Named Individuals: Has brother

```
Person(Fred) ^ hasSibling(Fred, ?s) ^ Man(?s)

→ hasBrother(Fred, ?s)
```

Example SWRL Rule with Literals and Built-ins: is adult?

Person(?p) ^ hasAge(?p,?age) ^
swrlb:greaterThan(?age,17)

→ Adult(?p)

Built-ins dramatically increase expressivity - most rules are not OWL syntactic sugar

Example SWRL Rule with String Built-ins

```
Person(?p) ^ hasNumber(?p, ?number) 
^ swrlb:startsWith(?number, "+") → 
hasInternationalNumber(?p, true)
```

Example SWRL Rule with Built-in Argument Binding

Person(?p) ^ hasSalaryInPounds(?p, ?pounds) ^ swrlb:multiply(?dollars, ?pounds, 2.0) → hasSalaryInDollars(?p, ?dollars)

Example SWRL Rule with Built-in Argument Binding II

Person(?p) ^ hasSalaryInPounds(?p, ?pounds) ^ swrlb:multiply(2.0, ?pounds, ?dollars) -> hasSalaryInDollars(?p, ?dollars)

Arguments can bind in any position - though generally an implementation will support binding of only the first argument

Can define new Built-in Libraries

Temporal built-ins:

- temporal:before("1999-11-01T10:00", "2000-02-01T11:12:12.000")
- temporal:duration(2, "1999-11-01", "2001-02-01", temporal:Years)


TBox built-ins:

- tbox:isDatatypeProperty(?p)
- tbox:isDirectSubPropertyOf(?sp, ?p)

Mathematical built-ins:

- swrlm:eval(?circumference, "2 * pi * r", ?r)

SWRLTab Built-in Libraries



Example SWRL Rule with OWL Class Expressions

(hasChild
$$\geq 1$$
)(?x) \rightarrow Parent(?x)

This does not say: all individuals with a child are parents

It says: all individuals that are members of an OWL class with the associated restriction that its hasChild property has a minimum cardinality of one

Example SWRL Rule with Inferred OWL Class Expressions

Parent(?x)
$$\rightarrow$$
 (hasChild >= 1)(?x)

Arbitrary OWL class expressions are allowed

Expression syntax may very, though Manchester Syntax common

SWRL Semantics

- Based on OWL-DL
- Has a formal semantics
- Complements OWL and fully semantically compatible
- More expressive yet at expense of decidability
- Use OWL if extra expressiveness not required (possible exception: querying)

OWL Class Expressions and the Open World Assumption

(hasChild
$$\geq 1$$
)(?x) \rightarrow Parent(?x)

This does not say: all individuals with a child are parents

It says: all individuals that are members of the OWL class with the associated restriction that its hasChild property has a minimum cardinality of one

Individuals with no known children may be classified as parents

SWRL and Open World Semantics: sameAs, differentFrom

Publication(?p) ^ hasAuthor(?p, ?y) ^ hasAuthor(?p, ?z) ^ differentFrom(?y, ?z) → cooperatedWith(?y, ?z)

Like OWL, SWRL does *not* adopt the unique name assumption Individuals must also be explicitly stated to be different (using, for example, owl:allDifferents restriction)

SWRL is Monotonic: does not Support Negated Atoms

Person(?p) ^ not hasCar(?p, ?c) → CarlessPerson(?p)

Not possible - language does not support negation here

Potential invalidation - what if a person later gets a car?

SWRL is Monotonic: retraction (or modification) not supported

Person(?p) ^ hasAge(?p,?age) ^ swrlb:add(?newage, ?age,1)

→ hasAge(?p, ?newage)

Incorrect - will run forever and attempt to assign an infinite number of values to hasAge property

Potential invalidation - essentially attempted retraction

SWRL is Monotonic: counting not supported

Publication(?p) ^ hasAuthor(?p,?a) ^ hasAuthor(?p,?a) ^ hasAuthor(?p,?a) ^ current-ontology

→ SingleAuthorPublication(?p)

Not expressible - open world applies

Potential invalidation - what if author is added later?

SWRL is Monotonic: counting not supported II

Publication(?p) ^ (hasAuthor = 1)(?p)


→ SingleAuthorPublication(?p)

Closure - though best expressed in OWL in this case

SWRLTab


- A Protégé-OWL development environment for working with SWRL rules
- Supports editing and execution of rules
- Extension mechanisms to work with thirdparty rule engines
- Mechanisms for users to define built-in method libraries
- Supports querying of ontologies


SWRLTab Wiki: http://protege.cim3.net/cgi-bin/wiki.pl?SWRLTab





What is the SWRL Editor?

- The SWRL Editor is an extension to Protégé-OWL that permits the interactive editing of SWRL rules.
- The editor can be used to create SWRL rules, edit existing SWRL rules, and read and write SWRL rules.
- It is accessible as a tab within Protégé-OWL.










Executing SWRL Rules

- SWRL is a language specification
- Well-defined semantics
- Developers must implement engine
- Or map to existing rule engines
- Hence, a bridge...

SWRL Rule Engine Bridge





SWRL Rule Engine Bridge


- Given an OWL knowledge base it will extract SWRL rules and relevant OWL knowledge.
- Also provides an API to assert inferred knowledge.
- Knowledge (and rules) are described in non Protégé-OWL API-specific way.
- These can then be mapped to a rule-engine specific rule and knowledge format.
- This mapping is developer's responsibility.


We used the SWRL Bridge to Integrate Jess Rule Engine with Protégé-OWL


- Jess is a Java-based rule engine.
- Jess system consists of a rule base, fact base, and an execution engine.
- Available free to academic users, for a small fee to non-academic users
- Has been used in Protégé-based tools, e.g., JessTab.











Outstanding Issues

- SWRL Bridge does not know about all OWL restrictions:
 - Contradictions with rules possible!
 - Consistency must be assured by the user incrementally running a reasoner.
 - Hard problem to solve in general.
- Integrated reasoner and rule engine would be ideal.
- Current solution with Pellet, though only with core built-in libraries.

SWRLTab Java APIs

- The SWRLTab provides APIs for all components
- These APIs are accessible to all OWL Protégé-OWL developers.
- Third party software can use these APIs to work directly with SWRL rules and integrate rules into their applications
- Fully documented in SWRLTab Wiki

SWRL and Querying

- SWRL is a rule language, not a query language
- However, a rule antecedent can be viewed as a pattern matching specification, i.e., a query
- With built-ins, language compliant query extensions are possible
- Hence: SQWRL (Semantic Query-Enhanced Web Rule Language; pronounced squirrel)

Example SQWRL Query

- Person(?p) ^ hasAge(?p,?age) ^ swrlb:greaterThan(?age,17)
 - → sqwrl:select(?p, ?age)

Ordering Query Results

- Person(?p) ^ hasAge(?p,?age) ^ swrlb:greaterThan(?age,17)
 - → sqwrl:select(?p, ?age) ^ sqwrl:orderBy(?age)

Counting Query Results

Person(?p) ^ hasCar(?p,?car)

→ sqwrl:select(?p) ^
sqwrl:count(?car)

Important: no way of asserting count in ontology!

Count all Owned Cars in Ontology

Person(?p) ^ hasCar(?p, ?c) → sqwrl:count(?c)

Count all Cars in Ontology

Car(?c) → sqwrl:count(?c)

Aggregation Queries: average age of persons in ontology

Person(?p) ^ hasAge(?p, ?age) -> sqwrl:avg(?age)

Also: sqwrl:max, sqwrl:min, sqwrl:sum

Queries and Rules Can Interact

```
Person(?p) ^ hasAge(?p,?age) 
^ swrlb:greaterThan(?age,17) 
→ Adult(?p)
```

Adult(?a) \rightarrow sqwrl:select(?a)

Example SWRL Query with OWL Restrictions

(hasChild >= 1)(?x) \rightarrow sqwrl:select(?x)

SQWRL can act as a DL query language

All Built-ins can be used in Queries

tbox:isDirectSubClassOf(?subClass, Person)

-> sqwrl:select(?subClass)


tbox:isSubPropertyOf(?supProperty, hasName)

-> sqwrl:select(?subProperty)

Note: use of property and class names as built-in arguments in not OWL DL


Important: these built-ins should be used in queries only – inference with them would definitely not be OWL DL


SQWRL Language FAQ




SQWRLTab

- Available as part of Protégé-OWL
 SWRLTab in current Protégé-3.4 beta
- Graphical interface to execute queries
- Low-level JDBC-like API for use in embedded applications
- Can use any existing rule engine back end







SWRLTab Wiki: http://protege.cim3.net/cgi-bin/wiki.pl?SWRLTab

